PROBLEM SET 1 22.S904 Nuclear Reactor Kinetics

Due: 17 September 2011

Bryan Herman

PART A

An 8-group point kinetics code was developed using Fortran. The code uses an xml-based user input where the user can input a reactivity and time profile. Also, each coarse time mesh can have an associated time step. The standard point kinetics equations (constant kinetics parameters, no external source) solved in the code are listed below:

$$\frac{d}{dt}T(t) = \frac{\rho(t) - \beta}{\Lambda}T(t) + \sum_{i} \lambda_{i}C_{i}(t)$$
$$\frac{d}{dt}C_{i}(t) = \frac{\beta_{i}}{\Lambda}T(t) - \lambda_{i}C_{i}(t).$$

Note that this gives us 9 equations, 1 for the power and 8 for the precursors. The steady state relationship between power and precursor can be determined by setting the time derivative in the precursor equation to zero. This yields

$$C_i^0 = \frac{\beta_i}{\Lambda \lambda_i} T^0.$$

In this code the power always starts from unity (fraction of nominal power) where the reactivity is initially zero. Setting up a linear system of these 9 equations yields

$$\frac{d}{dt} [N(t)] = \frac{d}{dt} \begin{bmatrix} T(t) \\ C_1(t) \\ C_2(t) \\ C_3(t) \\ C_3(t) \\ C_4(t) \\ C_5(t) \\ C_6(t) \\ C_7(t) \\ C_8(t) \end{bmatrix} = \begin{bmatrix} \frac{\rho(t) - \beta}{\Lambda} & \lambda_1 & \lambda_2 & \lambda_3 & \lambda_4 & \lambda_5 & \lambda_6 & \lambda_7 & \lambda_8 \\ \frac{\beta_1}{\Lambda} & -\lambda_1 & & & & & \\ \frac{\beta_2}{\Lambda} & & -\lambda_2 & & & & \\ \frac{\beta_3}{\Lambda} & & & -\lambda_3 & & & & \\ \frac{\beta_4}{\Lambda} & & & & -\lambda_4 & & & \\ \frac{\beta_5}{\Lambda} & & & & & -\lambda_5 & & \\ \frac{\beta_6}{\Lambda} & & & & & -\lambda_6 & & \\ \frac{\beta_7}{\Lambda} & & & & & & -\lambda_7 & \\ \frac{\beta_8}{\Lambda} & & & & & & -\lambda_8 \end{bmatrix} [N(t)] .$$

We can write this simply in matrix notation as

$$\frac{d}{dt}\left[N\left(t\right)\right] = \left[A\left(t\right)\right]\left[N\left(t\right)\right].$$

We can integrate this analytically as

$$[N(t)] = [N(0)] \exp \left\{ \int_0^{t'} [A(t')] dt' \right\}.$$

We can make an assumption that we take the matrix inside the integrate at time t if it does not rapidly change. Therefore, the final equation is

$$[N(t)] = [N(0)] \exp \{[A(t)] dt\}.$$

The exponential term on the right hand side is known as the matrix exponential. There are many ways to approximate this. In this Fortran implementation we use a Pade approximation. The solution algorithm is presented in the next section. We can discretize the system for any time step, k, as

$$\boxed{\left[N^{k}\right] = \left[N^{k-1}\right] \exp\left\{\left[A^{k}\right] dt^{k-1}\right\}.}$$

The source code to the main point kinetics is listed in the Appendix.

Solving the Matrix Exponential - Pade Approximation

The theory for the approximation of the Pade Approximation presented here is summarized from Nineteen Dubious Ways to Compute the Exponential of a Matrix, Twenty-Five Years Later. The Pade approximation of an exponential, $\exp(A)$ is defined as

$$R_{pq}(A) = [D_{pq}(A)]^{-1} E_{pq}(A).$$

The numerator and denominator are defined by

$$E_{pq}(A) = \sum_{j=0}^{q} \frac{(p+q-j)!p!}{(p+q)!j!(p-j)!} A^{j}$$

and

$$D_{pq}(A) = \sum_{j=0}^{q} \frac{(p+q-j)!p!}{(p+q)!j!(p-j)!} (-A^{j}).$$

Before the Pade approximation is made, the method of scaling and squaring is used to avoid numerical roundoff error. To scale the exponential, it can be written as

$$\exp(A) = \left[\exp(A/m)\right]^m.$$

The criterion to choose a parameter like m is that

$$||A/2^{e-1}||_{\infty} \le 1/2.$$

2

The integer value of e can be easily calculation and will be used at the end of the approximation to undo the scaling. After the matrix is scaled, the Pade approximation can be made. From above, the Pade approximation coefficients have numerous factorials and can be costly to compute. For example if we take this out to 4 terms and compute the diagonal approximate where, p = q = 6, we can see that it is

$$\sum_{j=0}^{q} \frac{(p+q-j)!p!}{(p+q)!j!(p-j)!} = 1 + \frac{11!6!}{12!5!} + \frac{10!6!}{12!2!4!} + \frac{9!6!}{12!3!3!} + \dots$$

We can simply some of the factorials as

$$1 + \frac{6}{12} + \frac{6 \cdot 5}{12 \cdot 11 \cdot 2} + \frac{6 \cdot 5 \cdot 4}{12 \cdot 11 \cdot 10 \cdot 3 \cdot 2} + \dots$$

What we see from this is that each succeeding term has the terms of the previous value when greater than the second term. Thus, we can always factor out all of the preceeding coefficients,

$$1 + \frac{6}{12} + \left(\frac{6}{12}\right) \frac{5}{11 \cdot 2} + \left(\frac{6}{12}\right) \left(\frac{5}{11 \cdot 2}\right) \frac{4}{10 \cdot 3} + \dots$$

Therefore for every term, we will now the multiplication of the terms in parathesis from the previous coefficient. We just need to write an algorithm for the terms in fractions. We notice that starting from j = 1 term, the value of the numerator has to do with q. The numerator can always be computed as num = q - j + 1. The denominator is always made up of two terms. One of them is the term number, j. The other is just 2q - j + 1. Therefore we can compute any Pade coefficient, c_j with the following algorithm:

$$c_j = c_{j-1} \frac{q-j+1}{j \cdot (2q-j+1)}, \quad j = 1, 2, \dots$$

Once the numerator, E_{qq} and denominator D_{qq} are computed the Pade approximation of them is just the division of the matrices. The last step is undo the scaling that was done at the beginning of the algorithm. This can be done with

$$\exp\left(A\right) \approx R_{qq}^{s}$$
.

The algorithm of this method is presented below. The acutal Fortran source code written is available in the Appendix. For all complicated matrix operations, LAPACK routines were used.

Algorithm 1 Matrix Exponential - Pade Approximation with Squaring and Scaling


- 1: compute norm-inf of matrix, $||A||_{\infty}$
- 2: get scaling parameter (s): determine e such that $norm = f \cdot 2^e$, $s = \max(0, e+1)$
- 3: initialize Pade numerator (E) and denominator (D) with a diagonal of ones (takes care of zeroth term)
- 4: initialize Pade coefficient, c=1
- 5: **for** j = 1, 2, 3, ...q **do**
- 6: Compute Pade coefficient, $c=c^*(q-j+1)/(j^*(2(q-j+1)))$
- 7: Perform matrix multiplication for A^{j}
- 8: Compute numerator, $E = E + c * A^j$
- 9: Compute denominator, $D = D(+/-)c * A^{j}$
- 10: end for
- 11: Compute approximate, $R = D^{-1}E$
- 12: Undo scaling, $E = E^s$

Part B


From the output of the code, the total value of beta and pnl is:

$$\beta = 0.006648$$
 $\Lambda = 1.866 \times 10^5 \,\mathrm{s.}$

The output of a point kinetics run can be viewed in the Appendix. The plot below shows fraction power vs. time during the transient for different time steps.



The plot above shows that once the time step becomes 0.1 seconds it very hard to visually see any differences. To be conservative we use a time step of 0.01 seconds. This time step is shown below with the corresponding reactivity input curve.



Part C

In this part we are tasked with writing an inverse point kinetics tool. In inverse kinetics, a power amplitude shape is used to determine a corresponding reactivity curve. We can start the derivation looking at the precursor balance equation for group i,

$$\frac{d}{dt}C_{i}\left(t\right) = \frac{\beta_{i}}{\Lambda}T\left(t\right) - \lambda C_{i}\left(t\right).$$

Since we know the power dependence the equations are no longer coupled. We can solve this differential equation analytically as

$$\frac{d}{dt}C_{i}(t) + \lambda C_{i}(t) = \frac{\beta_{i}}{\Lambda}T(t).$$

The integrating factor is $\exp(\lambda_i t)$. Going through the analysis we can arrive at:

$$\exp(\lambda_{i}t) \frac{d}{dt} C_{i}(t) + \exp(\lambda_{i}t) \lambda C_{i}(t) = \exp(\lambda_{i}t) \frac{\beta_{i}}{\Lambda} T(t),$$

$$\frac{d}{dt} \left[\exp(\lambda_{i}t) C_{i}(t) \right] = \exp(\lambda_{i}t) \frac{\beta_{i}}{\Lambda} T(t),$$

$$\int \frac{d}{dt} \left[\exp(\lambda_{i}t) C_{i}(t) \right] dt = \int \exp(\lambda_{i}t) \frac{\beta_{i}}{\Lambda} T(t) dt,$$

$$\exp(\lambda_{i}t) C_{i}(t) = \frac{\beta_{i}}{\Lambda} \int \exp(\lambda_{i}t) T(t) dt.$$

In order to perform that integral on the right hand side the analytic shape of the power amplitude must be known. However, if the time step is small enough we can just pull it out

of the integral as a constant and assume it is the average power of the time step. This is shown as

$$\exp(\lambda_{i}t) C_{i}(t) = \frac{\beta_{i}}{\Lambda} \bar{T} \int \exp(\lambda_{i}t) dt.$$

The general form of the equation is

$$\exp(\lambda_i t) C_i(t) = \frac{\beta_i}{\lambda_i \Lambda} \bar{T} \exp(\lambda_i t) + \tilde{C},$$

$$C_{i}(t) = \frac{\beta_{i}}{\lambda_{i}\Lambda}\bar{T} + \tilde{C}\exp(\lambda_{i}t).$$

The integration constant can be determine since we know from the initial power what the initial precursor concentration is. This constant can be evaluated as

$$C_i^0 = \frac{\beta_i}{\lambda_i \Lambda} \bar{T} + \tilde{C},$$

$$\tilde{C} = C_i^0 - \frac{\beta_i}{\lambda_i \Lambda} \bar{T}.$$

Substituting this back in to the general equation yields the final for of the precursor equation,

$$C_{i}(t) = \frac{\beta_{i}}{\lambda_{i}\Lambda}\bar{T} + \left(C_{i}^{0} - \frac{\beta_{i}}{\lambda_{i}\Lambda}\bar{T}\right)\exp\left(\lambda_{i}t\right),\,$$

$$C_{i}(t) = C_{i}^{0} \exp\left(\lambda_{i} t\right) + \frac{\beta_{i}}{\lambda_{i} \Lambda} \bar{T} \left[1 - \exp\left(\lambda_{i} t\right)\right].$$

This will be solved over a time step so that the above approximations are valid. The discretized equation then becomes

$$C_i^{k+1} = C_i^k \exp(\lambda_i dt^k) + \frac{\beta_i}{\lambda_i \Lambda} \bar{T}^k \left[1 - \exp(\lambda_i dt^k)\right].$$

The power equation solved for reactivity is

$$\rho(t) = \frac{\Lambda}{T(t)} \frac{d}{dt} T(t) + \beta - \frac{\Lambda}{T(t)} \sum_{i} \lambda_{i} C_{i}(t).$$

This can be discretized using a simple backward finite difference for the derivative so that

$$\rho^{k+1} = \frac{\Lambda}{T^{k+1}} \frac{T^{k+1} - T^k}{dt} + \beta - \frac{\Lambda}{T^{k+1}} \sum_i \lambda_i C_i^k.$$

These physics equation were implemented into a different Fortran code. The physics routines of the code can be viewed in the Appendix.


The given problem specifies a fixed power amplitude as a function of time as follows:

| $\int Time[s]$ | Power[W] |
|----------------|----------|
| 0 | 1 |
| 2 | 1 |
| 2.01 | 10000 |
| 10 | 10000 |
| 12 | 100 |
| 20 | 100 |


From running the code several times, we added the time step for each of these regions. To get the right shape after the reactivity insertion an extra time was put in to finer resolve the discretization:

| $\lceil Time \rceil$ | [s] Power [W] | Time step [s] |
|----------------------|---------------|--------------------|
| 0 | 1 | 0.1 |
| 2 | 1 | 3×10^{-8} |
| 2.01 | 10000 | 3×10^{-5} |
| 3.0 | 10000 | 0.001 |
| 10 | 10000 | 0.0001 |
| 12 | 100 | 0.1 |
| 20 | 100 | |


The input and output of the inverse point kinetics run is available in the Appendix. The resulting power and predicted reactivity curves are shown in the plot below.



Although the full scale of the reactivity can't be seen, the reactivity peaks at about 2700 dollars! This is obviously ridiculous for an actual reactor system. We can then take this reactivity curve and run it in the point kinetics solver to see if the original power amplitude shape can be recovered. The input and output of this run can be viewed in the Appendix. Since the data can get large to port between the codes, a binary output file is written and can be read in by point kinetics code. The same power and reactivity plot is shown below except from the point kinetics solver.



We can see from this plot that the power doesnt quite get up to 10000 W. Other than that it matches the plot from inverse kinetics well. For comparison we plot it against the reference power from the problem statement below.



We miss the power prediction by about 1.5%. This is due to the approximations of the time derivative of the power. As we make the time steps infinitesimally small, the power should be in better agreement. This was the best prediction with the amount of time needed to run the code.

APPENDIX

Source Code

Point Kinetics Main Physics Solver

```
module physics
!-module options
  implicit none
  private
  public :: run_kinetics
contains
! RUN_KINETICS
  subroutine run_kinetics()
!---external references
    use constants, only: NUM_PRECS
! use expokit, only: dense_pade
use global, only: pke, restart
use output, only: header
use solvers, only: expm_pade
!---local variables
    integer :: i ! loop counter
    real(8) :: dt ! local time step
!---begin execution
    ! print header for run
    call header("POINT KINETICS SIMULATION", level=1)
    ! set up coefficient matrix
    call setup_coefmat()
    ! set up initial conditions
    call set_init()
    ! begin loop through time steps
    do i = 1, sum(pke \% nt)
```

```
! print progress
      if (sum(pke\%nt) > 100000 .and. restart) then
       if (mod(i,sum(pke%nt)/10000) == 0) write(*,*) 'On Time Step: ',i
      end if
      ! compute exponential matrix
      call set_reactivity(i,dt)
     ! solve matrix exponential
      call dense pade(pke % coef, NUM PRECS + 1, dt, pke % expm) ! expokit
      call expm_pade(pke \% coef, NUM_PRECS + 1, dt, pke \% expm) ! my code
      ! get new vector
     pke \% N(:,i+1) = matmul(pke \% expm, pke \% N(:,i))
   end do
 end subroutine run kinetics
! SETUP COEFMAT
 subroutine setup_coefmat()
!---external references
   use constants, only: beta, NUM_PRECS, lambda, pnl
   use global,
                   only: pke
!---local variables
   integer :: i ! loop counter
!---begin execution
    ! set up coefficent matrix manually for time 0
   pke \% coef(1,1) = (pke \% rho(1)*sum(beta) - sum(beta))/pnl
    ! begin loop around rest of matrix
   do i = 2, NUM_PRECS + 1
     ! set row 1
     pke \% coef(1,i) = lambda(i - 1)
      ! set diagonal
```

```
pke \% coef(i,i) = -lambda(i - 1)
       ! set column 1
       \mathtt{pke} \ \% \ \mathtt{coef} (\mathtt{i} \ , 1) \ = \ \mathtt{beta} (\mathtt{i} \ -1) \ / \ \mathtt{pnl}
    end do
  end subroutine setup_coefmat
! SET INIT
  subroutine set_init()
!---external references
     {\tt use} \ {\tt constants} \ , \quad {\tt only} \colon \ {\tt ONE} \ , \ {\tt beta} \ , \ {\tt lambda} \ , \ {\tt pnl} \ , \ {\tt NUM\_PRECS}
    use global, only: pke
!---local variables
    integer :: i ! loop counter
!---begin execution
     ! set power at 1.0
    pke \% N(1,1) = ONE
     ! loop through precursors
    do i = 1, NUM_PRECS
       ! set initial value
       pke \% N(i+1,1) = beta(i)/(pnl*lambda(i)) * pke <math>\% N(1,1)
    end do
  end subroutine set_init
! SET REACTIVITY
  subroutine set_reactivity(i,dt)
!---external references
```

```
use constants, only: beta, pnl
    use global,
                       only: pke
   -arguments
    integer :: i ! current time step
    real(8) :: dt ! current dt
!---local variables
    integer :: idx ! interpolation index
!---begin execution
    ! check if index should be moved in input vectors
    if (i > sum(pke \% nt(1:pke \% idx))) pke \% idx = pke \% idx + 1
    idx = pke \% idx
    ! compute current time
    dt = pke \% dt(idx)
    \mathtt{pke} \ \% \ \mathtt{time}(\mathtt{i} + 1) = \mathtt{pke} \ \% \ \mathtt{time}(\mathtt{i}) + \mathtt{dt}
    ! interpolate on reactivity
     \verb|pke| \% \ \verb|react(i)| = \verb|pke| \% \ \verb|rho(idx)| + ((\verb|pke| \% \ \verb|rho(idx+1)| - \verb|pke| \% \ \verb|rho(idx)|) 
                                                (pke \% t(idx + 1) - pke \% t(idx)))
                                                (pke \% time(i+1) - pke \% t(idx))
    ! set values in coefficient matrix
    pke \% coef(1,1) = (pke \% react(i)*sum(beta) - sum(beta))/pnl
  end subroutine set_reactivity
end module physics
```

Inverse Point Kinetics Main Physics Solver

! RUN INVKINETICS subroutine run_invkinetics() ---external references use constants, only: NUM_PRECS, pnl, beta, lambda use global, only: ipke, total_time only: header use output, !---local variables integer :: i ! loop counter real(8) :: dt ! local time step real(8) :: avgpower ! avg. power between timesteps !---begin execution ! print header for run call header("INVERSE POINT KINETICS SIMULATION", level=1) ! set initial precursors ipke % N(1,1) = ipke % power(1) ipke % N(2:NUM_PRECS+1,1) = beta/(pnl*lambda) * ipke % N(1,1) ! begin loop through time steps do i = 1, sum(ipke % nt)! compute exponential matrix call set_power(i,dt) ! compute average power $avgpower = (ipke \% N(1,i) + ipke \% N(1,i+1)) / 2.0_8$! solve for precursors $\texttt{ipke} \hspace{0.2cm} \% \hspace{0.2cm} \texttt{N} \hspace{0.05cm} (\hspace{0.05cm} 2 : \texttt{NUM_PRECS} + 1, \texttt{i} \hspace{0.05cm} + \texttt{1}) \hspace{0.2cm} = \hspace{0.2cm} \texttt{ipke} \hspace{0.2cm} \% \hspace{0.2cm} \texttt{N} \hspace{0.05cm} (\hspace{0.05cm} 2 : \texttt{NUM_PRECS} + 1, \texttt{i} \hspace{0.05cm}) \hspace{0.2cm} * \hspace{0.2cm} \texttt{exp} \hspace{0.05cm} (-\hspace{0.05cm} \texttt{lambda*dt} \hspace{0.05cm}) \hspace{0.2cm} + \hspace{0.2cm} \& \hspace{0.2cm} (-\hspace{0.05cm} \texttt{lambda*lambda$ beta/(lambda*pnl)*(avgpower - avgpower*exp(-lambda*dt)) ! solve for reactivity ipke % react(i+1) = pnl/(ipke % N(1,i+1)) * & ((ipke % N(1,i+1) - ipke % N(1,i))/dt) +& & sum(beta) - pnl/ipke % N(1,i+1) * $sum(lambda*ipke \%N(2:NUM_PRECS+1,i+1))$! change to dollars ipke % react(i+1) = ipke % react(i+1)/sum(beta)

```
end do
 end subroutine run_invkinetics
! SET POWER
 subroutine set_power(i,dt)
  —external references
   use constants, only: beta, pnl
   use global,
                   only: ipke
!---arguments
   integer :: i ! current time step
   real(8) :: dt ! current dt
!---local variables
   integer :: idx ! interpolation index
!---begin execution
    ! check if index should be moved in input vectors
   if (i > sum(ipke \% nt(1:ipke \% idx))) ipke \% idx = ipke \% idx + 1
   idx = ipke \% idx
    ! compute current time
   dt = ipke \% dt(idx)
   ipke \% time(i+1) = ipke \% time(i) + dt
    ! interpolate on power
   ipke \% N(1,i+1) = ((ipke \% power(idx+1) - ipke \% power(idx)) / &
                      (ipke \% t(idx + 1) - ipke \% t(idx)))
                      (ipke \% time(i+1) - ipke \% t(idx)) + &
                      ipke % power(idx)
 end subroutine set_power
end module physics
```

Matrix Exponential Solver - Pade Approximation

```
module solvers
!-module options
 implicit none
 private
 public :: expm_pade
contains
! EXPM PADE
 subroutine expm_pade(A,N,dt,EXPM)
!---external references
   use constants, only: ZERO, ONE
                  only: norm_inf, ilog2
   use math,
!---arguments
                        ! dimension of square matrix
   integer :: N
   real(8) :: A(N,N)
                        ! the coefficient matrix
   real (8) :: dt
                        ! the time step
   real(8) :: EXPM(N,N) ! output matrix
!---local variables
   integer :: ex ! this is the integer exponent that goes into th scaling
                  ! this is the scaling parameter
   integer :: s
                  ! loop counter
   integer :: i
                  ! Pade order counter
   integer :: j
   integer :: q ! Pade approximation order
   integer :: info ! did LAPACK work?
   integer, allocatable :: IPIV(:) ! pivot elements from LU factorization
   real(8) :: norm ! the norm of the coefficient matrix for scaling
   real(8) :: c
                     ! Pade coefficient
   real(8) :: shift ! -1 or +1 and a function of J for D term
   real(8), allocatable :: At(:,:) ! the exponential being approximated
   real(8), allocatable :: X(:,:) ! accumulates multiplication of A
   real(8), allocatable :: Xt(:,:) ! temporary matrix for X
   real(8), allocatable :: E(:,:) ! numerator of Pade
   real(8), allocatable :: D(:,:) ! denominator of Pade
  -begin execution
```

```
! allocate temporary matrices
allocate(At(N,N))
allocate(X(N,N))
allocate(Xt(N,N))
allocate(E(N,N))
allocate(D(N,N))
allocate(IPIV(N))
! set matrix
At = A
! get matrix to be approximated in exponential
At = At * dt
! compute norm for scaling
norm = norm_inf(At,N)
! compute exponent integer exponent
ex = ilog2(norm)
! determine scaling parameter
s = max(0, ex+1)
! scale matrix by scaling parameter
\mathsf{At} = \mathsf{At} \ / \ (2 \! * \! * \! \mathsf{s})
! begin pade approximation for numerator and denominator of rational fun.
! Pade order 6, diagonal approximates
q = 6
! can easily set j=0 and j=1 parts right away
! begin loop to set the A^O terms in each E and D, not this is a diag of ones
E = ZERO
D = ZERO
doi=1,N
  E(i,i) = ONE
  D(i,i) = ONE
end do
c = ONE
c = 0.5_8! this is j=1 Pade coefficient
E = E + c*At! accomulate in sum
D = D - c*At! accumulate in sum
! begin rest of pade loop
X = At ! set A to X, X will accumulated the A^j in the Pade eq.
shift = ONE ! for order 2 term, the shift for denominator is +1
```

```
do j = 2,q
     ! compute Pade coefficient
     c = c * dble(q-j+1) / dble(j*(2*q-j+1))
     ! multiply A^{(j-1)} by A LAPACK ROUTINE
      call DGEMM('N','N',N,N,N,ONE,At,N,X,N,ZERO,Xt,N)
     ! move temporary to actual
     X = Xt
     ! accumulate into numerator and denominator
     E = E + c*X
     D = D + shift*c*X
     shift = -0NE
   end do
    ! compute rational function
   call DGETRF(N,N,D,N,IPIV,info)
! LU factorization LAPACK
   call DGETRS('N', N, N, D, N, IPIV, E, N, info) ! solves E = D^-1*E LAPACK
    ! perform squaring
   do i = 1,s
     call DGEMM('N','N',N,N,N,N,ONE,E,N,E,N,ZERO,Xt,N) ! does E*E LAPACK
     E = Xt
   end do
   ! set output matrix
   EXPM = E
   ! deallocate temporary matrices
   deallocate(At)
   deallocate(X)
   deallocate(Xt)
   deallocate(E)
   deallocate(D)
   deallocate(IPIV)
 end subroutine expm_pade
end module solvers
```

Part B Runs

Point Kinetics Input

Point Kinetics Output

Developed At: Massachusetts Institute of Technology

Version: 0.1

Git SHA1: 029eecfb04ec8c268d2584929167b6a17c1af2fc

Date/Time: 2012-09-15 11:33:19

Reading settings XML file...

INPUT SUMMARY

Number of time steps: 10000

Time (s) Rho (\$)
----0.000E+00 0.000E+00

| 5.000E+00 | 0.000E+00 |
|-----------|-----------|
| 7.000E+00 | 1.000E-01 |
| 2.700E+01 | 1.000E-01 |
| 2.900E+01 | 0.000E+00 |
| 1.000E+02 | 0.000E+00 |

PHYSICS SUMMARY

Number of Precursor Groups: 8

| beta | lambda |
|-----------|-----------|
| | |
| 2.180E-04 | 1.247E-02 |
| 1.023E-03 | 2.829E-02 |
| 6.050E-04 | 4.252E-02 |
| 1.310E-03 | 1.330E-01 |
| 2.200E-03 | 2.925E-01 |
| 6.000E-04 | 6.665E-01 |
| 5.400E-04 | 1.635E+00 |
| 1.520E-04 | 3.555E+00 |

Total Delayed Neutron Fraction 0.006648

Prompt Neutron Lifetime [s]: 1.866E-05

Plotting results with GNUPLOT...

SIMULATION SUMMARY

Total simulation time = 6.9000E-01 seconds

Simulation Finished.

Part C Runs


Inverse Point Kinetics Input

```
<?xml version="1.0" encoding="UTF-8"?>
<input>
```

```
<!-- Power Shape -->
<time> 0.0 2.0 2.01 3.0 10.0 12.0 20.0 </time>
<power> 1.0 1.0 10000.0 10000.0 10000.0 100.0 100.0 
<timestep> 0.1 3e-8 3e-5 1e-3 1e-4 0.1 </timestep>
```

Inverse Point Kinetics Output

</input>



Developed At: Massachusetts Institute of Technology

Version: 0.1

Git SHA1: 6ad3e02a086a0c13c2b75274e997addef77cc7e7

Date/Time: 2012-09-15 12:04:56

Reading settings XML file...

INPUT SUMMARY

Number of time steps: 393433

| Time (s) | Rho (\$) |
|-----------|-----------|
| | |
| 0.000E+00 | 1.000E+00 |
| 2.000E+00 | 1.000E+00 |
| 2.010E+00 | 1.000E+04 |
| 3.000E+00 | 1.000E+04 |
| 1.000E+01 | 1.000E+04 |
| 1.200E+01 | 1.000E+02 |
| 2.000E+01 | 1.000E+02 |

PHYSICS SUMMARY

Number of Precursor Groups: 8

| beta | lambda |
|-----------|-----------|
| | |
| 2.180E-04 | 1.247E-02 |
| 1.023E-03 | 2.829E-02 |
| 6.050E-04 | 4.252E-02 |
| 1.310E-03 | 1.330E-01 |
| 2.200E-03 | 2.925E-01 |
| 6.000E-04 | 6.665E-01 |
| 5.400E-04 | 1.635E+00 |
| 1.520E-04 | 3.555E+00 |

Plotting results with GNUPLOT...

SIMULATION SUMMARY

Total simulation time = 2.4000E-01 seconds

Simulation Finished.

Point Kinetics Input

```
<?xml version="1.0" encoding="UTF-8"?>
<input>
```

```
<restart> .TRUE. </restart>
```

</input>

Point Kinetics Output



| \ \ \ | \:\)\\ | /\:\\:_\ \ \\:\//\\::/ / \:\/\ |
|-------|--------|---------------------------------|
| _\/ | \\/\\/ | \\/ \\/ \\/ |

Developed At: Massachusetts Institute of Technology

Version: 0.1

Git SHA1: 029eecfb04ec8c268d2584929167b6a17c1af2fc

Date/Time: 2012-09-15 12:06:35

Reading settings XML file...

INPUT SUMMARY

Number of time steps: 393433

Restarted from inverse kinetics...

PHYSICS SUMMARY

Number of Precursor Groups: 8

| beta | lambda |
|-----------|-----------|
| | |
| 2.180E-04 | 1.247E-02 |
| 1.023E-03 | 2.829E-02 |
| 6.050E-04 | 4.252E-02 |
| 1.310E-03 | 1.330E-01 |
| 2.200E-03 | 2.925E-01 |
| 6.000E-04 | 6.665E-01 |
| 5.400E-04 | 1.635E+00 |
| 1.520E-04 | 3.555E+00 |

Total Delayed Neutron Fraction 0.006648

Prompt Neutron Lifetime [s]: 1.866E-05

| Plotting results with GNUPLOT | | |
|-------------------------------|---|--------------------|
| SIMULATION SUMMARY | | |
| Total simulation time | = | 4.8745E+02 seconds |
| Simulation Finished | | |